
[image: Buy Me A Coffee]
Erlang CQRS/Event Sourcing framework built on reckon-db.
[image: Architecture Overview]
Features
	Aggregate lifecycle with configurable TTL and passivation
	Per-event-type subscriptions (not per-stream)
	Command idempotency
	Middleware pipeline for command dispatch
	Event handlers with retry strategies and dead letter support
	Process managers (sagas) with compensation
	Projections with checkpointing
	Schema evolution via event upcasters
	Memory pressure monitoring with adaptive TTL
	Comprehensive telemetry integration

Installation
Add to your rebar.config:
{deps, [
    {evoq, "1.0.0"}
]}.
Quick Start
Defining an Aggregate
-module(bank_account).
-behaviour(evoq_aggregate).

-export([init/1, execute/2, apply/2]).

init(_AccountId) ->
    {ok, #{balance => 0, status => active}}.

execute(#{status := closed}, _Command) ->
    {error, account_closed};
execute(_State, #{command_type := open_account, initial_balance := B}) ->
    {ok, [#{event_type => <<"AccountOpened">>, data => #{balance => B}}]};
execute(#{balance := Bal}, #{command_type := deposit, amount := A}) ->
    {ok, [#{event_type => <<"MoneyDeposited">>, data => #{amount => A}}]}.

apply(State, #{event_type := <<"AccountOpened">>, data := #{balance := B}}) ->
    State#{balance => B};
apply(#{balance := B} = State, #{event_type := <<"MoneyDeposited">>, data := #{amount := A}}) ->
    State#{balance => B + A}.
Dispatching Commands
%% Create a command
Command = evoq_command:new(
    deposit,           %% command type
    bank_account,      %% aggregate module
    <<"acc-123">>,     %% aggregate id
    #{amount => 100}   %% payload
),

%% Dispatch it
{ok, Version, Events} = evoq_router:dispatch(Command).
Event Handlers
-module(notification_handler).
-behaviour(evoq_event_handler).

-export([interested_in/0, init/1, handle_event/4]).

interested_in() ->
    [<<"AccountOpened">>, <<"LargeDeposit">>].

init(_Config) ->
    {ok, #{}}.

handle_event(<<"AccountOpened">>, Event, _Metadata, State) ->
    send_welcome_email(Event),
    {ok, State};
handle_event(<<"LargeDeposit">>, Event, _Metadata, State) ->
    send_deposit_alert(Event),
    {ok, State}.
Process Managers (Sagas)
-module(order_fulfillment_pm).
-behaviour(evoq_process_manager).

-export([interested_in/0, correlate/2, handle/3, apply/2]).

interested_in() ->
    [<<"OrderPlaced">>, <<"PaymentReceived">>, <<"ItemShipped">>].

correlate(#{data := #{order_id := OrderId}}, _Meta) ->
    {continue, OrderId}.

handle(State, #{event_type := <<"OrderPlaced">>} = Event, _Meta) ->
    Cmd = evoq_command:new(process_payment, payment, OrderId, #{}),
    {ok, State, [Cmd]};
handle(State, #{event_type := <<"PaymentReceived">>}, _Meta) ->
    Cmd = evoq_command:new(ship_item, shipping, OrderId, #{}),
    {ok, State, [Cmd]};
handle(State, #{event_type := <<"ItemShipped">>}, _Meta) ->
    {ok, State#{status => completed}}.

apply(State, _Event) ->
    State.
Projections
-module(account_summary_projection).
-behaviour(evoq_projection).

-export([interested_in/0, init/1, project/4]).

interested_in() ->
    [<<"AccountOpened">>, <<"MoneyDeposited">>, <<"MoneyWithdrawn">>].

init(_Config) ->
    {ok, ReadModel} = evoq_read_model:new(evoq_read_model_ets, #{}),
    {ok, #{}, ReadModel}.

project(#{event_type := <<"AccountOpened">>, data := #{balance := B}},
        #{aggregate_id := Id}, State, ReadModel) ->
    {ok, NewRM} = evoq_read_model:put(Id, #{balance => B, tx_count => 0}, ReadModel),
    {ok, State, NewRM};
project(#{event_type := <<"MoneyDeposited">>, data := #{amount := A}},
        #{aggregate_id := Id}, State, ReadModel) ->
    {ok, Current} = evoq_read_model:get(Id, ReadModel),
    Updated = Current#{
        balance => maps:get(balance, Current) + A,
        tx_count => maps:get(tx_count, Current) + 1
    },
    {ok, NewRM} = evoq_read_model:put(Id, Updated, ReadModel),
    {ok, State, NewRM}.
Core Behaviors
Domain Artifacts
Domain artifacts stay inside the bounded context. They use atom keys and Erlang terms.
evoq_command
Commands represent intentions to change state. Formal contract for command modules.
-behaviour(evoq_command).

%% Required
-callback command_type() -> atom().
-callback new(Params :: map()) -> {ok, Command} | {error, Reason}.
-callback to_map(Command) -> map().

%% Optional
-callback validate(Command) -> ok | {ok, Command} | {error, Reason}.
-callback from_map(Map :: map()) -> {ok, Command} | {error, Reason}.
evoq_event
Events represent facts that have happened. Immutable once stored.
-behaviour(evoq_event).

%% Required
-callback event_type() -> atom().
-callback new(Params :: map()) -> Event.
-callback to_map(Event) -> map().

%% Optional
-callback from_map(Map :: map()) -> {ok, Event} | {error, Reason}.
Integration Artifacts
Integration artifacts cross bounded context boundaries. They use binary keys and are JSON-serializable.
evoq_fact
Facts translate domain events into payloads for external consumption via pg or mesh.
-behaviour(evoq_fact).

%% Required
-callback fact_type() -> binary().          %% e.g., <<"hecate.venture.initiated">>
-callback from_event(EventType :: atom(), EventData :: map(), Metadata :: map()) ->
    {ok, Payload :: map()} | skip.

%% Optional (defaults use OTP 27 json module)
-callback serialize(Payload :: map()) -> {ok, binary()} | {error, Reason}.
-callback deserialize(Binary :: binary()) -> {ok, map()} | {error, Reason}.
-callback schema() -> map().
evoq_hope
Hopes are outbound RPC requests between agents. Unlike facts (fire-and-forget), hopes expect a response.
-behaviour(evoq_hope).

%% Required
-callback hope_type() -> binary().
-callback new(Params :: map()) -> {ok, Hope} | {error, Reason}.
-callback to_payload(Hope) -> map().
-callback from_payload(Payload :: map()) -> {ok, Hope} | {error, Reason}.

%% Optional
-callback validate(Hope) -> ok | {error, Reason}.
See Artifacts Guide for detailed documentation and examples.
evoq_aggregate
Aggregates maintain business invariants and produce events.
-callback init(AggregateId :: binary()) -> {ok, State :: term()}.
-callback execute(State :: term(), Command :: map()) ->
    {ok, [Event :: map()]} | {error, Reason :: term()}.
-callback apply(State :: term(), Event :: map()) -> NewState :: term().

%% Optional: snapshotting
-callback snapshot(State :: term()) -> SnapshotData :: term().
-callback from_snapshot(SnapshotData :: term()) -> State :: term().
evoq_aggregate_lifespan
Controls aggregate lifecycle (TTL, passivation).
-callback after_event(Event :: map()) -> timeout() | infinity | hibernate | stop.
-callback after_command(Command :: map()) -> timeout() | infinity | hibernate | stop.
-callback after_error(Error :: term()) -> timeout() | infinity | hibernate | stop.
-callback on_timeout(State :: term()) -> {ok, action()} | {snapshot, action()}.
Default: 30-minute idle timeout, snapshot on passivation.
evoq_event_handler
Subscribe to events by type (not by stream).
-callback interested_in() -> [EventType :: binary()].
-callback init(Config :: map()) -> {ok, State :: term()}.
-callback handle_event(EventType, Event, Metadata, State) ->
    {ok, NewState} | {error, Reason}.
evoq_process_manager
Coordinate long-running business processes (sagas).
-callback interested_in() -> [EventType :: binary()].
-callback correlate(Event, Metadata) -> {start | continue | stop, ProcessId} | false.
-callback handle(State, Event, Metadata) -> {ok, State} | {ok, State, [Command]}.
-callback apply(State, Event) -> NewState.

%% Optional: saga compensation
-callback compensate(State, FailedCommand) -> {ok, [CompensatingCommand]} | skip.
evoq_projection
Build read models from events.
-callback interested_in() -> [EventType :: binary()].
-callback init(Config) -> {ok, State, ReadModel}.
-callback project(Event, Metadata, State, ReadModel) ->
    {ok, NewState, NewReadModel} | {skip, State, ReadModel}.
evoq_middleware
Intercept command dispatch.
-callback before_dispatch(Pipeline) -> {ok, Pipeline} | {error, Reason}.
-callback after_dispatch(Pipeline) -> {ok, Pipeline}.
-callback on_failure(Pipeline, Reason) -> {ok, Pipeline} | {error, Reason}.
Configuration
%% sys.config
[{evoq, [
    {store_id, my_store},

    {aggregate_defaults, #{
        idle_timeout => 1800000,    %% 30 minutes
        hibernate_after => 60000,   %% 1 minute
        snapshot_every => 100       %% events
    }},

    {aggregate_partitions, 4},

    {memory_monitor, #{
        check_interval => 10000,    %% 10 seconds
        elevated_threshold => 0.70,
        critical_threshold => 0.85
    }},

    {handler_defaults, #{
        consistency => eventual,
        start_from => origin
    }}
]}].
Memory Pressure Handling
The memory monitor adjusts aggregate TTLs based on system memory usage:
	Pressure Level	Memory Usage	TTL Factor
	normal	< 70%	1.0x
	elevated	70-85%	0.5x
	critical	> 85%	0.1x

Telemetry Events
All events follow the pattern: [evoq, component, action, stage]
Aggregate Events
	[evoq, aggregate, execute, start | stop | exception]

	[evoq, aggregate, init | hibernate | passivate | activate]

	[evoq, aggregate, snapshot, save | load]


Handler Events
	[evoq, handler, start | stop | exception]

	[evoq, handler, event, start | stop | exception]

	[evoq, handler, retry | dead_letter]


Process Manager Events
	[evoq, process_manager, start | stop]

	[evoq, process_manager, command | compensate]


Projection Events
	[evoq, projection, start | stop | exception]

	[evoq, projection, event | checkpoint]


Testing
# Unit tests
rebar3 eunit --dir=test/unit

# Integration tests
rebar3 ct

# Dialyzer
rebar3 dialyzer

# All tests with coverage
rebar3 do eunit, ct, cover

Key Design Decisions
Per-Event-Type Subscriptions
Unlike stream-based subscriptions, evoq subscribes by event type. This prevents subscription explosion when you have millions of aggregates.
Default TTL (Not Infinity!)
Aggregates have a 30-minute default idle timeout. This prevents unbounded memory growth that occurs with infinite lifespan defaults.
Partitioned Supervision
Aggregates are distributed across 4 partition supervisors using consistent hashing, preventing single-supervisor bottlenecks.
Documentation
Comprehensive guides are available:
	Architecture Overview - How the components work together
	Aggregates - Building domain models with event sourcing
	Event Handlers - Reacting to events with side effects
	Process Managers - Coordinating long-running workflows
	Projections - Building optimized read models
	Adapters - Integrating with different event stores

License
Apache-2.0


  

    Architecture

evoq implements CQRS (Command Query Responsibility Segregation) and Event Sourcing patterns for Erlang applications. This guide explains the core architecture and how the components work together.
[image: Architecture Overview]
Core Principles
1. Commands Produce Events, Events Update State
The fundamental flow in evoq:
	Command arrives (user intent)
	Aggregate validates and produces Events (facts)
	Events are persisted to the event store
	Events are applied to update aggregate state
	Events are routed to subscribers (handlers, projections, process managers)

This separation ensures:
	Commands can be rejected (validation failed)
	Events are immutable facts (already happened)
	State can be rebuilt by replaying events

2. Read Models Are Separate From Write Models
CQRS separates:
	Write Side	Read Side
	Commands	Queries
	Aggregates	Projections
	Event Store	Read Models
	Business logic	Optimized views

Benefits:
	Read models can be denormalized for fast queries
	Each read model optimized for specific use case
	Read models can be rebuilt from events

3. Event Type Subscriptions (Not Stream-Based)
Unlike stream-based subscriptions, evoq routes events by type:
%% Handler declares interest by event type
interested_in() ->
    [<<"OrderPlaced">>, <<"PaymentReceived">>].
Why this matters:
	1 million orders = 1 million streams
	Stream subscriptions = 1 million subscriptions (memory explosion)
	Type subscriptions = ~10 event types (constant memory)

Component Overview
evoq_dispatcher
Entry point for command processing. Routes commands through the middleware pipeline to the target aggregate.
%% Dispatch a command
{ok, Version, Events} = evoq_dispatcher:dispatch(Command).
evoq_middleware
Pluggable pipeline for command processing:
-behaviour(evoq_middleware).

before_dispatch(Pipeline) ->
    %% Validate, check idempotency, etc.
    {ok, Pipeline}.

after_dispatch(Pipeline) ->
    %% Post-processing
    {ok, Pipeline}.
Built-in middleware:
	Validation - Schema and business rule validation
	Idempotency - Prevent duplicate command processing
	Consistency - Strong or eventual consistency guarantees

evoq_aggregate
Business logic container. Validates commands and produces events.
-behaviour(evoq_aggregate).

%% Initialize state
init(AggregateId) -> {ok, InitialState}.

%% Validate and produce events
execute(State, Command) -> {ok, [Event]} | {error, Reason}.

%% Update state from event
apply(State, Event) -> NewState.
evoq_event_router
Routes persisted events to interested subscribers by event type.
Subscribers:
	Event Handlers (side effects)
	Process Managers (orchestration)
	Projections (read models)

evoq_event_handler
React to events with side effects:
-behaviour(evoq_event_handler).

interested_in() -> [<<"UserRegistered">>].

handle_event(EventType, Event, Metadata, State) ->
    send_welcome_email(Event),
    {ok, State}.
Features:
	Retry strategies (exponential backoff)
	Dead letter queue for failed events
	Checkpoint-based recovery

evoq_process_manager
Coordinate long-running business processes (sagas):
-behaviour(evoq_process_manager).

interested_in() -> [<<"OrderPlaced">>, <<"PaymentReceived">>].

correlate(Event, _Meta) ->
    {continue, maps:get(order_id, Event)}.

handle(State, Event, _Meta) ->
    Cmd = create_next_command(Event),
    {ok, State, [Cmd]}.

compensate(State, FailedCommand) ->
    {ok, [create_rollback_command(FailedCommand)]}.
evoq_projection
Build read models from events:
-behaviour(evoq_projection).

interested_in() -> [<<"OrderPlaced">>, <<"OrderShipped">>].

project(Event, Metadata, State, ReadModel) ->
    Updated = update_order_summary(Event, ReadModel),
    {ok, State, Updated}.
Data Flow
Command Processing
[image: Command Dispatch]
	Client creates command via evoq_command:new/4
	Dispatcher routes through middleware pipeline
	Aggregate executes command, produces events
	Events persisted to reckon-db
	Result returned to client

Event Distribution
[image: Event Routing]
	Events persisted to event store
	Event router receives notification
	Router looks up subscribers by event type
	Events delivered to matching handlers

Supervision Tree
evoq_sup
├── evoq_aggregates_sup           # Dynamic aggregate supervisors
│   ├── evoq_aggregate_partition_sup_0
│   │   └── evoq_aggregate (bank_account, "acc-1")
│   │   └── evoq_aggregate (bank_account, "acc-2")
│   ├── evoq_aggregate_partition_sup_1
│   │   └── ...
│   └── ... (4 partitions by default)
├── evoq_event_handler_sup        # Event handlers
│   └── evoq_event_handler (email_handler)
│   └── evoq_event_handler (audit_handler)
├── evoq_process_manager_sup      # Process managers
│   └── evoq_process_manager (order_pm, "order-123")
├── evoq_projection_sup           # Projections
│   └── evoq_projection (order_summary)
├── evoq_memory_monitor           # Memory pressure monitoring
└── evoq_event_router             # Event type registry
Partitioned Supervision
Aggregates are distributed across 4 partition supervisors using consistent hashing. This prevents:
	Single supervisor bottleneck
	Cascade failures affecting all aggregates
	Uneven load distribution

Integration with reckon-db
evoq uses reckon-db for event persistence:
%% Event store operations delegated to reckon-db
evoq_event_store:append(StoreId, StreamId, ExpectedVersion, Events)
evoq_event_store:read(StoreId, StreamId, StartVersion, Count)
evoq_event_store:subscribe(StoreId, EventTypes, Handler)
The adapter pattern allows different backends:
%% Configure adapter in sys.config
{evoq, [
    {event_store_adapter, reckon_evoq}
]}
Domain vs Integration Artifacts
evoq formalizes the distinction between internal and external data:
		Domain Artifacts	Integration Artifacts
	Types	Commands, Events	Facts, Hopes
	Scope	Inside bounded context	Cross boundaries
	Keys	Atoms	Binaries (JSON-safe)
	Transport	In-process, event store	pg, mesh, RPC

Domain events are implementation details. Integration facts are explicit public contracts. A process manager decides which domain events become integration facts, transforming internal structure into stable external schemas.
Domain Event (internal)           Integration Fact (external)
  venture_initiated_v1       -->    <<"hecate.venture.initiated">>
  (atom keys, full state)          (binary keys, curated subset)
See the Artifacts Guide for detailed documentation and examples.
Next Steps
	Artifacts Guide - Domain and integration artifact behaviours
	Aggregates Guide - Deep dive into aggregate patterns
	Event Handlers Guide - Building reactive systems
	Process Managers Guide - Orchestrating workflows
	Projections Guide - Building read models



  

    Aggregates

Aggregates are the core building blocks of your domain. They encapsulate business logic, maintain invariants, and produce events that represent state changes.
[image: Aggregate Lifecycle]
The Aggregate Pattern
An aggregate in evoq:
	Receives commands - User intents to change state
	Validates business rules - Ensures invariants are maintained
	Produces events - Facts representing what happened
	Applies events - Updates internal state

-module(bank_account).
-behaviour(evoq_aggregate).

-export([init/1, execute/2, apply/2]).

%% Initialize a new aggregate
init(_AccountId) ->
    {ok, #{balance => 0, status => active}}.

%% Validate command and produce events
execute(#{status := closed}, _Command) ->
    {error, account_closed};
execute(#{balance := Balance}, #{command_type := withdraw, amount := Amount})
  when Amount > Balance ->
    {error, insufficient_funds};
execute(_State, #{command_type := deposit, amount := Amount}) ->
    {ok, [#{event_type => <<"MoneyDeposited">>, data => #{amount => Amount}}]}.

%% Apply event to update state
apply(#{balance := Balance} = State, #{event_type := <<"MoneyDeposited">>, data := #{amount := Amount}}) ->
    State#{balance => Balance + Amount}.
Required Callbacks
init/1
Called when an aggregate is first created or reactivated.
-spec init(AggregateId :: binary()) -> {ok, State :: term()}.

init(AccountId) ->
    %% Return initial state for new aggregate
    {ok, #{
        account_id => AccountId,
        balance => 0,
        opened_at => undefined,
        status => new
    }}.
execute/2
Validates the command against current state and produces events.
-spec execute(State :: term(), Command :: map()) ->
    {ok, [Event :: map()]} | {error, Reason :: term()}.

execute(#{status := new}, #{command_type := open_account, initial_deposit := Amount}) ->
    {ok, [
        #{event_type => <<"AccountOpened">>, data => #{initial_deposit => Amount}},
        #{event_type => <<"MoneyDeposited">>, data => #{amount => Amount}}
    ]};
execute(#{status := active, balance := Balance}, #{command_type := withdraw, amount := Amount})
  when Amount =< Balance ->
    {ok, [#{event_type => <<"MoneyWithdrawn">>, data => #{amount => Amount}}]};
execute(#{status := active, balance := Balance}, #{command_type := withdraw, amount := Amount})
  when Amount > Balance ->
    {error, {insufficient_funds, Balance, Amount}}.
Key patterns:
	Use pattern matching on state for preconditions
	Use guards for numeric validations
	Return {ok, [Events]} on success
	Return {error, Reason} on validation failure
	Never produce events if validation fails

apply/2
Updates state from a single event. Called for each event produced by execute/2 and during replay.
-spec apply(State :: term(), Event :: map()) -> NewState :: term().

apply(State, #{event_type := <<"AccountOpened">>, data := Data}) ->
    State#{
        status => active,
        opened_at => erlang:system_time(millisecond),
        initial_deposit => maps:get(initial_deposit, Data)
    };
apply(#{balance := Balance} = State, #{event_type := <<"MoneyDeposited">>, data := #{amount := Amount}}) ->
    State#{balance => Balance + Amount};
apply(#{balance := Balance} = State, #{event_type := <<"MoneyWithdrawn">>, data := #{amount := Amount}}) ->
    State#{balance => Balance - Amount}.
Key patterns:
	Must be pure (no side effects)
	Must be deterministic (same input = same output)
	Handle all event types the aggregate produces
	Called during replay - don't assume order

Optional Callbacks
snapshot/1 and from_snapshot/1
Enable snapshotting for faster aggregate loading:
-spec snapshot(State :: term()) -> SnapshotData :: term().
snapshot(State) ->
    %% Serialize state for storage
    State.

-spec from_snapshot(SnapshotData :: term()) -> State :: term().
from_snapshot(SnapshotData) ->
    %% Deserialize state from storage
    SnapshotData.
When loading an aggregate:
	Load latest snapshot (if exists)
	Replay events after snapshot version
	Much faster than replaying all events

Lifecycle Management
Aggregates are not kept in memory forever. The evoq_aggregate_lifespan behavior controls:
	How long to keep aggregate active
	When to hibernate (reduce memory)
	When to passivate (stop process)
	When to save snapshots

Default Lifespan
%% Default behavior
-module(evoq_aggregate_lifespan_default).

after_event(_Event) ->
    30 * 60 * 1000.  %% 30 minute timeout

after_command(_Command) ->
    30 * 60 * 1000.  %% 30 minute timeout

after_error(_Error) ->
    infinity.  %% Keep active on error for debugging

on_timeout(State) ->
    {snapshot, stop}.  %% Save snapshot, then stop
Custom Lifespan
-module(my_aggregate_lifespan).
-behaviour(evoq_aggregate_lifespan).

after_event(#{event_type := <<"HighValueTransaction">>}) ->
    infinity;  %% Keep high-value accounts active
after_event(_) ->
    5 * 60 * 1000.  %% 5 minute timeout for others

after_command(_) ->
    10 * 60 * 1000.

after_error({insufficient_funds, _, _}) ->
    60 * 1000;  %% 1 minute for expected errors
after_error(_) ->
    infinity.  %% Keep active for unexpected errors

on_timeout(State) ->
    case maps:get(balance, State, 0) > 10000 of
        true -> {snapshot, hibernate};  %% Snapshot high-value, hibernate
        false -> {ok, stop}  %% Just stop low-value
    end.
Memory Pressure
The evoq_memory_monitor adjusts aggregate TTLs based on system memory:
	Pressure	Memory Usage	TTL Factor	Effect
	normal	< 70%	1.0x	Normal TTL
	elevated	70-85%	0.5x	Faster passivation
	critical	> 85%	0.1x	Aggressive cleanup

This prevents OOM conditions under load while keeping aggregates active when memory is available.
Dispatching Commands
Create and dispatch commands:
%% Create a command
Command = evoq_command:new(
    deposit,           %% command type atom
    bank_account,      %% aggregate module
    <<"acc-123">>,     %% aggregate id
    #{amount => 100}   %% payload
),

%% With metadata
CommandWithMeta = evoq_command:with_metadata(Command, #{
    user_id => <<"user-456">>,
    correlation_id => <<"corr-789">>
}),

%% Dispatch
case evoq_dispatcher:dispatch(CommandWithMeta) of
    {ok, NewVersion, ProducedEvents} ->
        %% Success - version and events returned
        ok;
    {error, Reason} ->
        %% Command rejected by aggregate or middleware
        handle_error(Reason)
end.
Concurrency Control
reckon-db uses optimistic concurrency with expected versions:
%% First write to new stream
append(StoreId, StreamId, -1, Events)  %% -1 = stream must not exist

%% Append to existing stream
append(StoreId, StreamId, 5, Events)   %% Must be at version 5

%% Append without version check
append(StoreId, StreamId, -2, Events)  %% -2 = any version
If another process appended events, you get:
{error, {wrong_expected_version, Expected, Actual}}
evoq handles this automatically with retries when appropriate.
Testing Aggregates
Test aggregates in isolation:
-module(bank_account_tests).
-include_lib("eunit/include/eunit.hrl").

deposit_test() ->
    {ok, State0} = bank_account:init(<<"test-acc">>),

    %% Execute command
    Command = #{command_type => deposit, amount => 100},
    {ok, Events} = bank_account:execute(State0, Command),

    %% Apply events
    State1 = lists:foldl(fun bank_account:apply/2, State0, Events),

    ?assertEqual(100, maps:get(balance, State1)).

insufficient_funds_test() ->
    {ok, State0} = bank_account:init(<<"test-acc">>),

    Command = #{command_type => withdraw, amount => 100},
    ?assertEqual({error, insufficient_funds}, bank_account:execute(State0, Command)).
Best Practices
1. Keep Aggregates Small
Aggregates should be cohesive:
	One aggregate = one consistency boundary
	If two things can change independently, they're separate aggregates
	Large aggregates = more conflicts, slower loading

2. Use Domain Language
Events should reflect business meaning:
%% Good - business language
#{event_type => <<"AccountOverdrawn">>}
#{event_type => <<"LoyaltyPointsEarned">>}

%% Bad - CRUD language
#{event_type => <<"AccountUpdated">>}
#{event_type => <<"BalanceChanged">>}
3. Events Are Facts
Events represent what happened, not what will happen:
%% Good - past tense, immutable fact
<<"MoneyDeposited">>
<<"AccountClosed">>

%% Bad - future/present tense
<<"DepositMoney">>
<<"CloseAccount">>
4. Validate Early
Reject invalid commands before producing events:
execute(_, #{command_type := deposit, amount := Amount}) when Amount =< 0 ->
    {error, {invalid_amount, Amount}};
execute(State, #{command_type := deposit, amount := Amount}) ->
    %% Only valid deposits reach here
    {ok, [#{event_type => <<"MoneyDeposited">>, data => #{amount => Amount}}]}.
Next Steps
	Event Handlers - React to events
	Process Managers - Coordinate workflows
	Projections - Build read models



  

    Event Handlers

Event handlers react to domain events with side effects. They enable loose coupling between the core domain and external concerns like notifications, integrations, and analytics.
[image: Event Routing]
When to Use Event Handlers
Use event handlers for:
	Notifications - Send emails, SMS, push notifications
	External integrations - Update external systems, APIs
	Analytics - Track metrics, log events
	Side effects - Anything that doesn't affect domain state

Do not use event handlers for:
	Updating aggregate state (use apply/2)
	Building read models (use projections)
	Coordinating workflows (use process managers)

Basic Event Handler
-module(email_notification_handler).
-behaviour(evoq_event_handler).

-export([interested_in/0, init/1, handle_event/4]).

%% Declare which event types to receive
interested_in() ->
    [<<"UserRegistered">>, <<"PasswordReset">>, <<"OrderShipped">>].

%% Initialize handler state
init(_Config) ->
    {ok, #{email_service => email_client:connect()}}.

%% Handle each event
handle_event(<<"UserRegistered">>, Event, _Metadata, State) ->
    #{email := Email, name := Name} = maps:get(data, Event),
    send_welcome_email(Email, Name, State),
    {ok, State};

handle_event(<<"PasswordReset">>, Event, _Metadata, State) ->
    #{email := Email, token := Token} = maps:get(data, Event),
    send_password_reset_email(Email, Token, State),
    {ok, State};

handle_event(<<"OrderShipped">>, Event, Metadata, State) ->
    #{order_id := OrderId, tracking := Tracking} = maps:get(data, Event),
    CustomerEmail = lookup_customer_email(OrderId),
    send_shipping_notification(CustomerEmail, Tracking, State),
    {ok, State}.
Required Callbacks
interested_in/0
Declare which event types this handler wants to receive:
-spec interested_in() -> [EventType :: binary()].

interested_in() ->
    [<<"OrderPlaced">>, <<"OrderCancelled">>].
The event router only delivers matching events, reducing overhead.
init/1
Initialize handler state on startup:
-spec init(Config :: map()) -> {ok, State :: term()} | {error, Reason :: term()}.

init(Config) ->
    ApiKey = maps:get(api_key, Config),
    Client = external_service:connect(ApiKey),
    {ok, #{client => Client, retry_count => 0}}.
handle_event/4
Process a single event:
-spec handle_event(EventType :: binary(), Event :: map(), Metadata :: map(), State :: term()) ->
    {ok, NewState :: term()} | {error, Reason :: term()}.

handle_event(EventType, Event, Metadata, State) ->
    %% EventType: <<"OrderPlaced">>
    %% Event: #{event_type => ..., data => ..., ...}
    %% Metadata: #{aggregate_id => ..., version => ..., timestamp => ...}
    %% State: Handler's internal state

    do_side_effect(Event),
    {ok, State}.
Retry Strategies
Event handlers can fail (network issues, service unavailable). evoq supports retry strategies:
-module(my_handler).
-behaviour(evoq_event_handler).

-export([interested_in/0, init/1, handle_event/4, retry_strategy/0]).

%% Optional callback for retry configuration
retry_strategy() ->
    #{
        max_retries => 5,
        initial_delay => 1000,      %% 1 second
        max_delay => 60000,         %% 1 minute max
        backoff => exponential,     %% exponential | linear | constant
        jitter => true              %% Add randomness to prevent thundering herd
    }.

handle_event(EventType, Event, _Meta, State) ->
    case call_external_api(Event) of
        {ok, _} -> {ok, State};
        {error, temporary} -> {error, temporary};  %% Will retry
        {error, permanent} -> {ok, State}          %% Don't retry, mark as handled
    end.
Dead Letter Queue
Events that fail after all retries go to the dead letter queue:
%% Handle dead letters
evoq_dead_letter:list(HandlerName) -> [FailedEvent].
evoq_dead_letter:retry(HandlerName, EventId) -> ok | {error, Reason}.
evoq_dead_letter:discard(HandlerName, EventId) -> ok.

%% Monitor dead letters
evoq_dead_letter:count(HandlerName) -> non_neg_integer().
Dead letters include:
	Original event
	Handler name
	Failure reason
	Number of attempts
	Timestamps

Consistency Modes
Event handlers support two consistency modes:
Eventual Consistency (Default)
Events processed asynchronously. Best for most use cases.
init(Config) ->
    {ok, #{}, #{consistency => eventual}}.
Properties:
	Non-blocking command dispatch
	Handler may lag behind writes
	Failures don't affect command success

Strong Consistency
Events processed before command returns. Use sparingly.
init(Config) ->
    {ok, #{}, #{consistency => strong}}.
Properties:
	Blocks until handler completes
	Handler failure fails the command
	Higher latency
	Use only when necessary (audit, compliance)

Start From Position
Control where handler starts processing:
init(Config) ->
    {ok, #{}, #{
        start_from => origin    %% Process all historical events
        %% start_from => current   %% Only new events
        %% start_from => {position, 1000}  %% From specific position
    }}.
Idempotency
Handlers may receive the same event multiple times (restarts, redelivery). Make handlers idempotent:
handle_event(<<"OrderShipped">>, Event, Metadata, State) ->
    EventId = maps:get(event_id, Event),
    case already_processed(EventId, State) of
        true ->
            %% Already handled, skip
            {ok, State};
        false ->
            send_shipping_notification(Event),
            {ok, mark_processed(EventId, State)}
    end.
Or use external idempotency keys:
handle_event(<<"PaymentProcessed">>, Event, _Meta, State) ->
    IdempotencyKey = maps:get(event_id, Event),
    case payment_gateway:charge(Event, #{idempotency_key => IdempotencyKey}) of
        {ok, _} -> {ok, State};
        {error, already_processed} -> {ok, State};  %% Gateway handled idempotency
        {error, Reason} -> {error, Reason}
    end.
Multiple Handlers Per Event
Different handlers can process the same event type:
%% Handler 1: Send email
-module(email_handler).
interested_in() -> [<<"OrderPlaced">>].

%% Handler 2: Update analytics
-module(analytics_handler).
interested_in() -> [<<"OrderPlaced">>].

%% Handler 3: Notify warehouse
-module(warehouse_handler).
interested_in() -> [<<"OrderPlaced">>].
All three receive OrderPlaced events independently.
Handler Registration
Register handlers in your application supervisor:
-module(my_app_sup).

init([]) ->
    Children = [
        %% Register event handlers
        {email_handler, {evoq_event_handler, start_link, [email_handler, #{}]},
            permanent, 5000, worker, [evoq_event_handler]},

        {analytics_handler, {evoq_event_handler, start_link, [analytics_handler, #{
            api_key => <<"...">>
        }]}, permanent, 5000, worker, [evoq_event_handler]}
    ],
    {ok, {{one_for_one, 5, 10}, Children}}.
Testing Event Handlers
Test handlers in isolation:
-module(email_handler_tests).
-include_lib("eunit/include/eunit.hrl").

handle_user_registered_test() ->
    %% Setup
    {ok, State} = email_handler:init(#{email_service => mock_email}),

    Event = #{
        event_type => <<"UserRegistered">>,
        data => #{email => <<"test@example.com">>, name => <<"Test User">>}
    },
    Metadata = #{aggregate_id => <<"user-123">>, version => 1},

    %% Execute
    {ok, _NewState} = email_handler:handle_event(<<"UserRegistered">>, Event, Metadata, State),

    %% Verify email was sent (via mock)
    ?assert(mock_email:was_called_with(<<"test@example.com">>)).
Telemetry Events
Event handlers emit telemetry:
	Event	Measurements	Metadata
	[evoq, handler, start]	system_time	handler, event_type
	[evoq, handler, stop]	duration	handler, event_type
	[evoq, handler, exception]	duration	handler, error, stacktrace
	[evoq, handler, retry]	attempt	handler, event_type, reason
	[evoq, handler, dead_letter]	system_time	handler, event_type, reason

Best Practices
1. Keep Handlers Focused
One handler = one responsibility:
%% Good - focused
-module(order_email_handler).      %% Just emails
-module(order_analytics_handler).   %% Just analytics

%% Bad - doing too much
-module(order_handler).  %% Emails + analytics + logging + ...
2. Handle Failures Gracefully
handle_event(EventType, Event, _Meta, State) ->
    try
        do_risky_operation(Event),
        {ok, State}
    catch
        error:temporary_failure ->
            {error, temporary};  %% Retry
        error:permanent_failure ->
            log_and_alert(Event),
            {ok, State}  %% Don't retry, but continue
    end.
3. Log Extensively
handle_event(EventType, Event, Metadata, State) ->
    logger:info("Processing ~p for aggregate ~p",
        [EventType, maps:get(aggregate_id, Metadata)]),

    Result = do_work(Event),

    logger:info("Completed ~p: ~p", [EventType, Result]),
    {ok, State}.
4. Monitor Dead Letters
Set up alerts for dead letter growth:
%% In your monitoring system
check_dead_letters() ->
    Handlers = [email_handler, analytics_handler],
    lists:foreach(fun(Handler) ->
        Count = evoq_dead_letter:count(Handler),
        telemetry:execute([my_app, dead_letters], #{count => Count}, #{handler => Handler})
    end, Handlers).
Next Steps
	Process Managers - Coordinate multi-step workflows
	Projections - Build read models
	Architecture - System overview



  

    Process Managers

Process managers (also called sagas) coordinate long-running business processes that span multiple aggregates. They react to events and dispatch commands to drive workflows forward.
[image: Process Manager Flow]
When to Use Process Managers
Use process managers when:
	A business process spans multiple aggregates
	You need to coordinate a sequence of operations
	Failures require compensation (rollback)
	The process has state that persists across events

Examples:
	Order fulfillment (payment → inventory → shipping)
	User onboarding (account → profile → welcome email)
	Money transfer (debit source → credit destination)

Basic Process Manager
-module(order_fulfillment_pm).
-behaviour(evoq_process_manager).

-export([interested_in/0, correlate/2, handle/3, apply/2]).

%% Events this process manager reacts to
interested_in() ->
    [<<"OrderPlaced">>, <<"PaymentReceived">>, <<"InventoryReserved">>, <<"ItemShipped">>].

%% Route events to the correct process instance
correlate(#{data := #{order_id := OrderId}}, _Metadata) ->
    {continue, OrderId}.

%% React to events by dispatching commands
handle(State, #{event_type := <<"OrderPlaced">>} = Event, _Meta) ->
    OrderId = maps:get(order_id, maps:get(data, Event)),
    Amount = maps:get(amount, maps:get(data, Event)),

    %% Dispatch command to payment aggregate
    Cmd = evoq_command:new(process_payment, payment, OrderId, #{amount => Amount}),
    {ok, State#{status => awaiting_payment}, [Cmd]};

handle(State, #{event_type := <<"PaymentReceived">>}, _Meta) ->
    OrderId = maps:get(order_id, State),

    %% Dispatch command to inventory aggregate
    Cmd = evoq_command:new(reserve_inventory, inventory, OrderId, #{}),
    {ok, State#{status => awaiting_inventory}, [Cmd]};

handle(State, #{event_type := <<"InventoryReserved">>}, _Meta) ->
    OrderId = maps:get(order_id, State),

    %% Dispatch command to shipping aggregate
    Cmd = evoq_command:new(ship_item, shipping, OrderId, #{}),
    {ok, State#{status => awaiting_shipment}, [Cmd]};

handle(State, #{event_type := <<"ItemShipped">>}, _Meta) ->
    %% Process complete
    {ok, State#{status => completed}}.

%% Update process state from events
apply(State, #{event_type := <<"OrderPlaced">>, data := Data}) ->
    State#{
        order_id => maps:get(order_id, Data),
        customer_id => maps:get(customer_id, Data),
        items => maps:get(items, Data),
        status => started
    };
apply(State, _Event) ->
    State.
Required Callbacks
interested_in/0
Declare which event types this process manager reacts to:
-spec interested_in() -> [EventType :: binary()].

interested_in() ->
    [<<"OrderPlaced">>, <<"PaymentFailed">>, <<"InventoryUnavailable">>].
correlate/2
Route events to the correct process instance:
-spec correlate(Event :: map(), Metadata :: map()) ->
    {start, ProcessId :: term()} |
    {continue, ProcessId :: term()} |
    {stop, ProcessId :: term()} |
    false.

%% Start a new process
correlate(#{event_type := <<"OrderPlaced">>, data := #{order_id := OrderId}}, _Meta) ->
    {start, OrderId};

%% Continue existing process
correlate(#{event_type := <<"PaymentReceived">>, data := #{order_id := OrderId}}, _Meta) ->
    {continue, OrderId};

%% Stop the process
correlate(#{event_type := <<"OrderCancelled">>, data := #{order_id := OrderId}}, _Meta) ->
    {stop, OrderId};

%% Ignore event (no matching process)
correlate(_, _) ->
    false.
handle/3
React to events and optionally dispatch commands:
-spec handle(State :: term(), Event :: map(), Metadata :: map()) ->
    {ok, NewState :: term()} |
    {ok, NewState :: term(), Commands :: [map()]}.

%% Just update state
handle(State, #{event_type := <<"OrderPlaced">>}, _Meta) ->
    {ok, State#{started_at => erlang:system_time()}};

%% Update state and dispatch commands
handle(State, #{event_type := <<"PaymentReceived">>}, _Meta) ->
    Cmd1 = evoq_command:new(reserve_inventory, inventory, OrderId, #{}),
    Cmd2 = evoq_command:new(notify_warehouse, warehouse, OrderId, #{}),
    {ok, State#{payment_received => true}, [Cmd1, Cmd2]}.
apply/2
Update process state from events (called before handle/3):
-spec apply(State :: term(), Event :: map()) -> NewState :: term().

apply(State, #{event_type := <<"OrderPlaced">>, data := Data}) ->
    State#{
        order_id => maps:get(order_id, Data),
        total => maps:get(total, Data)
    };
apply(State, #{event_type := <<"PaymentReceived">>, data := #{amount := Amount}}) ->
    State#{paid_amount => Amount}.
Compensation (Rollback)
When a step fails, the process manager can compensate by undoing previous steps:
-module(money_transfer_pm).
-behaviour(evoq_process_manager).

-export([interested_in/0, correlate/2, handle/3, apply/2, compensate/2]).

interested_in() ->
    [<<"TransferInitiated">>, <<"SourceDebited">>, <<"DestinationCreditFailed">>].

handle(State, #{event_type := <<"TransferInitiated">>}, _Meta) ->
    %% Step 1: Debit source account
    Cmd = evoq_command:new(debit, account, SourceId, #{amount => Amount}),
    {ok, State#{status => debiting_source}, [Cmd]};

handle(State, #{event_type := <<"SourceDebited">>}, _Meta) ->
    %% Step 2: Credit destination account
    Cmd = evoq_command:new(credit, account, DestId, #{amount => Amount}),
    {ok, State#{status => crediting_dest}, [Cmd]};

handle(State, #{event_type := <<"DestinationCreditFailed">>}, _Meta) ->
    %% Credit failed - need to compensate
    {ok, State#{status => compensating}}.

%% Compensation callback
-spec compensate(State :: term(), FailedCommand :: map()) ->
    {ok, CompensatingCommands :: [map()]} | skip.

compensate(#{source_id := SourceId, amount := Amount}, #{command_type := credit}) ->
    %% Credit failed, refund the source account
    RefundCmd = evoq_command:new(credit, account, SourceId, #{
        amount => Amount,
        reason => <<"transfer_failed">>
    }),
    {ok, [RefundCmd]};

compensate(_, _) ->
    skip.
State Machine Pattern
Process managers naturally model state machines:
-module(order_state_machine_pm).

%% State transitions
handle(#{status := new} = State, #{event_type := <<"OrderPlaced">>}, _) ->
    {ok, State#{status => awaiting_payment}, [process_payment_cmd()]};

handle(#{status := awaiting_payment} = State, #{event_type := <<"PaymentReceived">>}, _) ->
    {ok, State#{status => awaiting_shipment}, [ship_order_cmd()]};

handle(#{status := awaiting_payment} = State, #{event_type := <<"PaymentFailed">>}, _) ->
    {ok, State#{status => cancelled}, [notify_customer_cmd()]};

handle(#{status := awaiting_shipment} = State, #{event_type := <<"ItemShipped">>}, _) ->
    {ok, State#{status => completed}};

%% Invalid transition - ignore
handle(State, _Event, _Meta) ->
    {ok, State}.
Timeout Handling
Handle timeouts for long-running processes:
-module(booking_pm).
-behaviour(evoq_process_manager).

-export([interested_in/0, correlate/2, handle/3, apply/2, timeout/0]).

timeout() ->
    15 * 60 * 1000.  %% 15 minute timeout

handle(#{status := awaiting_confirmation, started_at := Started} = State, timeout, _Meta) ->
    %% Timeout triggered - cancel the booking
    Cmd = evoq_command:new(cancel_booking, booking, BookingId, #{
        reason => timeout
    }),
    {ok, State#{status => timed_out}, [Cmd]}.
Correlation Strategies
By Entity ID
Most common - route by the main entity:
correlate(#{data := #{order_id := OrderId}}, _) ->
    {continue, OrderId}.
By Correlation ID
Use metadata for cross-aggregate correlation:
correlate(_Event, #{correlation_id := CorrelationId}) ->
    {continue, CorrelationId}.
Composite Key
When multiple entities involved:
correlate(#{data := #{source := Src, dest := Dst}}, _) ->
    {continue, {transfer, Src, Dst}}.
Testing Process Managers
Test the state machine in isolation:
-module(order_pm_tests).
-include_lib("eunit/include/eunit.hrl").

full_workflow_test() ->
    %% Initial state
    State0 = #{},

    %% Order placed
    {start, OrderId} = order_pm:correlate(order_placed_event(), #{}),
    State1 = order_pm:apply(State0, order_placed_event()),
    {ok, State2, [PaymentCmd]} = order_pm:handle(State1, order_placed_event(), #{}),

    ?assertEqual(awaiting_payment, maps:get(status, State2)),
    ?assertEqual(process_payment, maps:get(command_type, PaymentCmd)),

    %% Payment received
    State3 = order_pm:apply(State2, payment_received_event()),
    {ok, State4, [ShipCmd]} = order_pm:handle(State3, payment_received_event(), #{}),

    ?assertEqual(awaiting_shipment, maps:get(status, State4)),
    ?assertEqual(ship_item, maps:get(command_type, ShipCmd)),

    %% Item shipped
    State5 = order_pm:apply(State4, item_shipped_event()),
    {ok, State6} = order_pm:handle(State5, item_shipped_event(), #{}),

    ?assertEqual(completed, maps:get(status, State6)).
Telemetry Events
Process managers emit telemetry:
	Event	Measurements	Metadata
	[evoq, process_manager, start]	system_time	name, process_id
	[evoq, process_manager, stop]	duration	name, process_id, final_state
	[evoq, process_manager, command]	system_time	name, command_type
	[evoq, process_manager, compensate]	system_time	name, failed_command

Best Practices
1. Keep Processes Short-Lived
Long-running processes accumulate state and risk:
	Design for completion in minutes/hours, not days
	Split long workflows into smaller processes
	Use timeouts to handle stuck processes

2. Make Steps Idempotent
Commands may be dispatched multiple times:
handle(State, Event, _Meta) ->
    case already_dispatched(Event, State) of
        true -> {ok, State};
        false ->
            Cmd = create_command(Event),
            {ok, mark_dispatched(Event, State), [Cmd]}
    end.
3. Handle All Failure Modes
interested_in() ->
    [
        %% Happy path
        <<"OrderPlaced">>, <<"PaymentReceived">>, <<"ItemShipped">>,
        %% Failure cases
        <<"PaymentFailed">>, <<"InventoryUnavailable">>, <<"ShippingFailed">>
    ].
4. Log State Transitions
handle(State, Event, _Meta) ->
    OldStatus = maps:get(status, State),
    {ok, NewState, Cmds} = do_handle(State, Event),
    NewStatus = maps:get(status, NewState),

    logger:info("Process ~p: ~p -> ~p on ~p",
        [maps:get(process_id, State), OldStatus, NewStatus, maps:get(event_type, Event)]),

    {ok, NewState, Cmds}.
Next Steps
	Projections - Build read models
	Event Handlers - Side effects
	Architecture - System overview



  

    Projections

Projections build read models from events. They transform the event stream into optimized views for queries, enabling fast reads without impacting write performance.
[image: Projection Flow]
When to Use Projections
Use projections for:
	Query-optimized views - Denormalized data for fast reads
	Aggregated data - Counts, sums, statistics
	Cross-aggregate views - Data from multiple aggregates
	Search indexes - Full-text search, filtering

The key insight: read models can be rebuilt by replaying events.
Understanding Event Envelope Structure
IMPORTANT: Projections receive events wrapped in an evoq_event envelope. The first parameter to project/4 is a map with this structure:
#{
    event_type => <<"OrderPlaced">>,     %% Event type (binary)
    data => #{                            %% ← YOUR BUSINESS EVENT PAYLOAD
        order_id => <<"ord-123">>,
        customer_id => <<"cust-456">>,
        total => 99.99,
        items => [...]
    },
    metadata => #{                        %% ← Envelope metadata
        aggregate_id => <<"ord-123">>,
        correlation_id => <<"req-abc">>,
        causation_id => <<"cmd-xyz">>,
        timestamp => 1703001234567
    },
    tags => [<<"tenant:acme">>],
    ...
}
Key fields:
	event_type - Binary identifying the event type
	data - Your business event payload (extract this)
	metadata - Cross-cutting metadata (correlation_id, causation_id, etc.)
	tags - Optional tags for cross-stream queries

The second parameter (Metadata) is the envelope metadata, duplicated for convenience.
See Event Envelope Guide for complete details.
Basic Projection
-module(order_summary_projection).
-behaviour(evoq_projection).

-export([interested_in/0, init/1, project/4]).

%% Events this projection processes
interested_in() ->
    [<<"OrderPlaced">>, <<"OrderShipped">>, <<"OrderCancelled">>].

%% Initialize projection with a read model backend
init(_Config) ->
    {ok, ReadModel} = evoq_read_model:new(evoq_read_model_ets, #{}),
    {ok, #{}, ReadModel}.

%% Project each event into the read model
project(#{event_type := <<"OrderPlaced">>, data := Data}, Metadata, State, ReadModel) ->
    OrderId = maps:get(aggregate_id, Metadata),
    Summary = #{
        order_id => OrderId,
        customer_id => maps:get(customer_id, Data),
        total => maps:get(total, Data),
        items => maps:get(items, Data),
        status => placed,
        placed_at => maps:get(timestamp, Metadata)
    },
    {ok, NewRM} = evoq_read_model:put(OrderId, Summary, ReadModel),
    {ok, State, NewRM};

project(#{event_type := <<"OrderShipped">>, data := Data}, Metadata, State, ReadModel) ->
    OrderId = maps:get(aggregate_id, Metadata),
    case evoq_read_model:get(OrderId, ReadModel) of
        {ok, Summary} ->
            Updated = Summary#{
                status => shipped,
                tracking_number => maps:get(tracking_number, Data),
                shipped_at => maps:get(timestamp, Metadata)
            },
            {ok, NewRM} = evoq_read_model:put(OrderId, Updated, ReadModel),
            {ok, State, NewRM};
        {error, not_found} ->
            %% Order not found - skip
            {skip, State, ReadModel}
    end;

project(#{event_type := <<"OrderCancelled">>}, Metadata, State, ReadModel) ->
    OrderId = maps:get(aggregate_id, Metadata),
    case evoq_read_model:get(OrderId, ReadModel) of
        {ok, Summary} ->
            Updated = Summary#{
                status => cancelled,
                cancelled_at => maps:get(timestamp, Metadata)
            },
            {ok, NewRM} = evoq_read_model:put(OrderId, Updated, ReadModel),
            {ok, State, NewRM};
        {error, not_found} ->
            {skip, State, ReadModel}
    end.
Required Callbacks
interested_in/0
Declare which event types to process:
-spec interested_in() -> [EventType :: binary()].

interested_in() ->
    [<<"UserRegistered">>, <<"ProfileUpdated">>, <<"UserDeactivated">>].
init/1
Initialize projection state and read model:
-spec init(Config :: map()) ->
    {ok, State :: term(), ReadModel :: term()} |
    {error, Reason :: term()}.

init(Config) ->
    %% Choose read model backend
    Backend = maps:get(backend, Config, evoq_read_model_ets),
    {ok, ReadModel} = evoq_read_model:new(Backend, Config),

    %% Initial state
    State = #{
        events_processed => 0,
        last_event_time => undefined
    },

    {ok, State, ReadModel}.
project/4
Transform an event into read model updates:
-spec project(Event :: map(), Metadata :: map(), State :: term(), ReadModel :: term()) ->
    {ok, NewState :: term(), NewReadModel :: term()} |
    {skip, State :: term(), ReadModel :: term()}.

project(Event, Metadata, State, ReadModel) ->
    %% Event: #{event_type => ..., data => ...}
    %% Metadata: #{aggregate_id => ..., version => ..., timestamp => ...}

    %% Update read model
    {ok, NewRM} = update_read_model(Event, ReadModel),

    %% Update projection state
    NewState = State#{
        events_processed => maps:get(events_processed, State) + 1,
        last_event_time => maps:get(timestamp, Metadata)
    },

    {ok, NewState, NewRM}.
Read Model Backends
evoq provides a evoq_read_model abstraction with multiple backends:
ETS (In-Memory)
Fast, volatile storage. Good for caching and development.
init(_Config) ->
    {ok, RM} = evoq_read_model:new(evoq_read_model_ets, #{
        table_name => order_summaries
    }),
    {ok, #{}, RM}.
PostgreSQL
Persistent, queryable storage. Good for production.
init(Config) ->
    {ok, RM} = evoq_read_model:new(evoq_read_model_postgres, #{
        pool => my_pool,
        table => <<"order_summaries">>
    }),
    {ok, #{}, RM}.
Custom Backend
Implement the evoq_read_model behavior:
-module(my_custom_read_model).
-behaviour(evoq_read_model).

-export([new/1, get/2, put/3, delete/2, list/1]).

new(Config) -> {ok, State}.
get(Key, State) -> {ok, Value} | {error, not_found}.
put(Key, Value, State) -> {ok, NewState}.
delete(Key, State) -> {ok, NewState}.
list(State) -> {ok, [Key]}.
Checkpointing
Projections checkpoint their position in the event stream:
%% Automatic checkpointing (default: every 100 events)
init(Config) ->
    {ok, RM} = evoq_read_model:new(evoq_read_model_ets, #{}),
    {ok, #{}, RM, #{checkpoint_interval => 100}}.

%% Manual checkpointing
project(Event, Metadata, State, ReadModel) ->
    %% Do work...

    %% Checkpoint after expensive operations
    case should_checkpoint(State) of
        true ->
            evoq_projection:checkpoint(self()),
            {ok, reset_checkpoint_state(State), NewRM};
        false ->
            {ok, State, NewRM}
    end.
On restart, projections resume from their last checkpoint.
Rebuilding Projections
Read models can be rebuilt from events:
%% Rebuild a specific projection
evoq_projection:rebuild(order_summary_projection).

%% Rebuild with options
evoq_projection:rebuild(order_summary_projection, #{
    from => origin,        %% Start from beginning
    batch_size => 1000,    %% Process in batches
    parallel => 4          %% Use 4 workers
}).
This is powerful:
	Fix bugs in projection logic
	Add new fields to read model
	Migrate to a new storage backend
	Test projection changes safely

Cross-Aggregate Projections
Projections can combine data from multiple aggregates:
-module(customer_orders_projection).
-behaviour(evoq_projection).

interested_in() ->
    [<<"CustomerRegistered">>, <<"OrderPlaced">>, <<"OrderShipped">>].

project(#{event_type := <<"CustomerRegistered">>, data := Data}, Meta, State, RM) ->
    CustomerId = maps:get(aggregate_id, Meta),
    Customer = #{
        customer_id => CustomerId,
        name => maps:get(name, Data),
        email => maps:get(email, Data),
        orders => [],
        total_spent => 0
    },
    {ok, NewRM} = evoq_read_model:put(CustomerId, Customer, RM),
    {ok, State, NewRM};

project(#{event_type := <<"OrderPlaced">>, data := Data}, _Meta, State, RM) ->
    CustomerId = maps:get(customer_id, Data),
    case evoq_read_model:get(CustomerId, RM) of
        {ok, Customer} ->
            OrderSummary = #{
                order_id => maps:get(order_id, Data),
                total => maps:get(total, Data),
                status => placed
            },
            Updated = Customer#{
                orders => [OrderSummary | maps:get(orders, Customer)],
                total_spent => maps:get(total_spent, Customer) + maps:get(total, Data)
            },
            {ok, NewRM} = evoq_read_model:put(CustomerId, Updated, RM),
            {ok, State, NewRM};
        {error, not_found} ->
            {skip, State, RM}
    end.
Aggregation Projections
Build statistics and summaries:
-module(daily_stats_projection).
-behaviour(evoq_projection).

interested_in() ->
    [<<"OrderPlaced">>, <<"OrderCancelled">>].

project(#{event_type := <<"OrderPlaced">>, data := Data}, Meta, State, RM) ->
    Date = date_from_timestamp(maps:get(timestamp, Meta)),

    Stats = case evoq_read_model:get(Date, RM) of
        {ok, Existing} -> Existing;
        {error, not_found} -> #{orders => 0, revenue => 0, cancelled => 0}
    end,

    Updated = Stats#{
        orders => maps:get(orders, Stats) + 1,
        revenue => maps:get(revenue, Stats) + maps:get(total, Data)
    },

    {ok, NewRM} = evoq_read_model:put(Date, Updated, RM),
    {ok, State, NewRM}.
Querying Read Models
Query projections directly:
%% Get single item
{ok, OrderSummary} = evoq_read_model:get(OrderId, ReadModel).

%% List all items
{ok, AllOrders} = evoq_read_model:list(ReadModel).

%% Backend-specific queries (PostgreSQL)
{ok, RecentOrders} = evoq_read_model_postgres:query(
    ReadModel,
    "SELECT * FROM order_summaries WHERE status = $1 ORDER BY placed_at DESC LIMIT $2",
    [<<"placed">>, 10]
).
Testing Projections
Test projections with event sequences:
-module(order_summary_projection_tests).
-include_lib("eunit/include/eunit.hrl").

project_order_lifecycle_test() ->
    %% Setup
    {ok, State0, RM0} = order_summary_projection:init(#{}),

    %% Order placed
    PlacedEvent = #{
        event_type => <<"OrderPlaced">>,
        data => #{customer_id => <<"cust-1">>, total => 100, items => []}
    },
    PlacedMeta = #{aggregate_id => <<"order-1">>, timestamp => 1000},

    {ok, State1, RM1} = order_summary_projection:project(PlacedEvent, PlacedMeta, State0, RM0),

    %% Verify
    {ok, Summary1} = evoq_read_model:get(<<"order-1">>, RM1),
    ?assertEqual(placed, maps:get(status, Summary1)),
    ?assertEqual(100, maps:get(total, Summary1)),

    %% Order shipped
    ShippedEvent = #{
        event_type => <<"OrderShipped">>,
        data => #{tracking_number => <<"TRACK123">>}
    },
    ShippedMeta = #{aggregate_id => <<"order-1">>, timestamp => 2000},

    {ok, _State2, RM2} = order_summary_projection:project(ShippedEvent, ShippedMeta, State1, RM1),

    %% Verify
    {ok, Summary2} = evoq_read_model:get(<<"order-1">>, RM2),
    ?assertEqual(shipped, maps:get(status, Summary2)),
    ?assertEqual(<<"TRACK123">>, maps:get(tracking_number, Summary2)).
Telemetry Events
Projections emit telemetry:
	Event	Measurements	Metadata
	[evoq, projection, start]	system_time	name
	[evoq, projection, stop]	duration	name
	[evoq, projection, event]	duration	name, event_type
	[evoq, projection, checkpoint]	position	name
	[evoq, projection, rebuild, start]	system_time	name
	[evoq, projection, rebuild, stop]	duration, events_processed	name

Best Practices
1. Design for Queries
Think about how the data will be queried:
%% Good - optimized for "get orders by customer"
project(Event, Meta, State, RM) ->
    CustomerId = maps:get(customer_id, Event),
    Key = {customer, CustomerId},
    %% One lookup per customer
    ...
2. Handle Missing Data
Events may arrive out of order or reference missing entities:
project(Event, _Meta, State, RM) ->
    case evoq_read_model:get(ParentId, RM) of
        {ok, Parent} ->
            %% Update normally
            {ok, State, NewRM};
        {error, not_found} ->
            %% Parent doesn't exist - skip or create placeholder
            {skip, State, RM}
    end.
3. Keep Projections Idempotent
Projections may replay events during rebuild:
project(Event, Meta, State, RM) ->
    EventId = maps:get(event_id, Event),
    case already_projected(EventId, State) of
        true -> {skip, State, RM};
        false ->
            %% Project normally
            NewState = mark_projected(EventId, State),
            {ok, NewState, NewRM}
    end.
4. Monitor Lag
Track how far behind projections are:
project(Event, Meta, State, RM) ->
    EventTime = maps:get(timestamp, Meta),
    Lag = erlang:system_time(millisecond) - EventTime,

    telemetry:execute([my_app, projection, lag], #{lag_ms => Lag}, #{
        projection => ?MODULE
    }),

    %% Continue projecting...
    {ok, State, NewRM}.
5. Version Read Models
Support schema evolution:
project(Event, Meta, State, RM) ->
    {ok, Existing} = evoq_read_model:get(Key, RM),

    %% Migrate old format
    Migrated = case Existing of
        #{version := 1} -> migrate_v1_to_v2(Existing);
        #{version := 2} -> Existing;
        _ -> #{version => 2}  %% New record
    end,

    Updated = update_record(Migrated, Event),
    {ok, NewRM} = evoq_read_model:put(Key, Updated#{version => 2}, RM),
    {ok, State, NewRM}.
Next Steps
	Event Handlers - Side effects
	Process Managers - Orchestration
	Architecture - System overview



  

    Adapters

evoq uses an adapter pattern to integrate with different event store backends. This guide explains how to configure and implement adapters.
Overview
evoq doesn't include any specific event store implementation. Instead, it defines adapter behaviors that must be implemented to connect to your chosen backend.
┌─────────────────┐     ┌─────────────────┐     ┌─────────────────┐
│    evoq     │────>│   Adapter       │────>│  Event Store    │
│  (Framework)    │     │ (Interface)     │     │   (Backend)     │
└─────────────────┘     └─────────────────┘     └─────────────────┘
Available Adapters
reckon-db-gater (Recommended)
The evoq_esdb_gater_adapter connects to reckon-db via the gateway protocol.
Add to your dependencies:
{deps, [
    {evoq, "0.2.0"},
    {reckon_db_gater, "0.4.3"}
]}.
Configure in sys.config:
{evoq, [
    {event_store_adapter, evoq_esdb_gater_adapter},
    {store_id, my_store}
]}
Adapter Behaviors
evoq defines three adapter behaviors:
evoq_adapter (Event Store)
Core event operations:
-callback append(StoreId, StreamId, ExpectedVersion, Events) -> {ok, Version} | {error, Reason}.
-callback read(StoreId, StreamId, StartVersion, Count, Direction) -> {ok, Events} | {error, Reason}.
-callback read_all(StoreId, StreamId, Direction) -> {ok, Events} | {error, Reason}.
-callback version(StoreId, StreamId) -> integer().
-callback exists(StoreId, StreamId) -> boolean().
-callback list_streams(StoreId) -> {ok, [StreamId]} | {error, Reason}.
-callback delete_stream(StoreId, StreamId) -> ok | {error, Reason}.
evoq_snapshot_adapter
Snapshot operations for aggregate optimization:
-callback save(StoreId, StreamId, Version, Data, Metadata) -> ok | {error, Reason}.
-callback read(StoreId, StreamId) -> {ok, Snapshot} | {error, not_found}.
-callback read_at_version(StoreId, StreamId, Version) -> {ok, Snapshot} | {error, not_found}.
-callback delete(StoreId, StreamId) -> ok | {error, Reason}.
-callback list_versions(StoreId, StreamId) -> {ok, [Version]} | {error, Reason}.
evoq_subscription_adapter
Subscription operations for event delivery:
-callback subscribe(StoreId, Type, Selector, Name, Opts) -> {ok, SubscriptionId} | {error, Reason}.
-callback unsubscribe(StoreId, SubscriptionId) -> ok | {error, Reason}.
-callback ack(StoreId, Name, StreamId, Position) -> ok | {error, Reason}.
-callback get_checkpoint(StoreId, Name) -> {ok, Position} | {error, not_found}.
-callback list(StoreId) -> {ok, [Subscription]} | {error, Reason}.
Implementing a Custom Adapter
To implement a custom adapter:
	Create a module that implements evoq_adapter (required)
	Optionally implement evoq_snapshot_adapter and evoq_subscription_adapter
	Configure evoq to use your adapter

Example skeleton:
-module(my_custom_adapter).
-behaviour(evoq_adapter).
-behaviour(evoq_snapshot_adapter).
-behaviour(evoq_subscription_adapter).

%% evoq_adapter callbacks
-export([append/4, read/5, read_all/3, version/2, exists/2, list_streams/1, delete_stream/2]).

%% evoq_snapshot_adapter callbacks
-export([save/5, read/2, read_at_version/3, delete/2, list_versions/2]).

%% evoq_subscription_adapter callbacks
-export([subscribe/5, unsubscribe/2, ack/4, get_checkpoint/2, list/1]).

%% Implementation...
append(StoreId, StreamId, ExpectedVersion, Events) ->
    %% Connect to your event store and append events
    {ok, NewVersion}.

%% ... implement other callbacks
Configuration
Full adapter configuration options:
{evoq, [
    %% Required: which adapter to use
    {event_store_adapter, evoq_esdb_gater_adapter},

    %% Store identifier (passed to adapter)
    {store_id, my_store},

    %% Optional: separate adapters for snapshots/subscriptions
    {snapshot_adapter, evoq_esdb_gater_snapshot_adapter},
    {subscription_adapter, evoq_esdb_gater_subscription_adapter}
]}
Local Development
For local development, use _checkouts to link adapters:
mkdir -p _checkouts
ln -s /path/to/reckon-db-gater _checkouts/reckon_db_gater

This allows modifying the adapter code alongside your application.
Testing with Mocks
Use meck to mock adapters in tests:
my_test() ->
    meck:new(evoq_esdb_gater_adapter, [passthrough]),
    meck:expect(evoq_esdb_gater_adapter, append, fun(_, _, _, Events) ->
        {ok, length(Events)}
    end),

    %% Run your test
    ok = my_module:do_something(),

    meck:unload(evoq_esdb_gater_adapter).
See Also
	Architecture - How adapters fit into the overall system
	reckon-db-gater - Gateway adapter for reckon-db
	reckon-db - BEAM-native event store

